

! List of variables in the data file (NAMES = ) and list of used variables (USEVARIABLES =)

VARIABLE:

NAMES =

SNAR1 SNAR2 SNAR3 PANAR1 PANAR2 PANAR3 FRNAR1 FRNAR2 FRNAR3

SMAC1 SMAC2 SMAC3 PAMAC1 PAMAC2 PAMAC3 FRMAC1 FRMAC2 FRMAC3

SPSY1 SPSY2 SPSY3 PAPSY1 PAPSY2 PAPSY3 FRPSY1 FRPSY2 FRPSY3;

USEVARIABLES =

SNAR1 SNAR2 SNAR3 PANAR1 PANAR2 PANAR3 FRNAR1 FRNAR2 FRNAR3

SMAC1 SMAC2 SMAC3 PAMAC1 PAMAC2 PAMAC3 FRMAC1 FRMAC2 FRMAC3

SPSY1 SPSY2 SPSY3 PAPSY1 PAPSY2 PAPSY3 FRPSY1 FRPSY2 FRPSY3;

ANALYSIS: ! TECHNICAL SPECIFICATION FOR THE ANALYSIS

TYPE = GENERAL; ! standard CFA / SEM specification

MODEL: ! Specification of the statistical model (here CTC(M-1))

! definition of the three indicator-specific trait-variables for Narcissism

NAR1 by SNAR1 PANAR1 FRNAR1;

NAR2 by SNAR2 PANAR2 FRNAR2;

NAR3 by SNAR3 PANAR3 FRNAR3;

! method-factor for parent-report of Narcissism

MFPANAR by PANAR1 PANAR2 PANAR3;

! method-factor for friend-report of Narcissism

MFFRNAR by FRNAR1 FRNAR2 FRNAR3;

! Restriction of inadmissible correlations to 0

NAR1 NAR2 NAR3 WITH MFPANAR@0 MFFRNAR@0;

! Specification of Machiavellianism (trait-variable)

MACH by SMAC1 SMAC2 SMAC3 PAMAC1 PAMAC2 PAMAC3 FRMAC1 FRMAC2 FRMAC3;

! method-factor for parent-report of Machiavellianism

MFPAMACH by PAMAC1 PAMAC2 PAMAC3;

! method-factor for friend-report of Machiavellianism

MFFRMACH by FRMAC1 FRMAC2 FRMAC3;

! Restriction of inadmissible correlations to 0

MACH with MFPAMACH@0 mFFRMACH@0;

! Specification of Psychopathy (trait-variable)

PSY by SPSY1 SPSY2 SPSY3 PAPSY1 PAPSY2 PAPSY3 FRPSY1 FRPSY2 FRPSY3;

! method-factor for parent-report of Psychopathy

MFPAPSY by PAPSY1 PAPSY2 PAPSY3;

! method-factor for friend-report of Psychopathy

MFFRPSY by FRPSY1 FRPSY2 FRPSY3;

! Restriction of inadmissible correlations to 0

PSY with MFPAPSY@0 MFFRPSY@0;

! Output options calling for standardized model parameters

OUTPUT: STDYX;